

## Simulazione del decadimento radioattivo

*Eeguire un programma di calcolatore somiglia molto a effettuare un esperimento. [...] L'impiego del calcolatore estende così i confini della scienza sperimentale, consentendoci di effettuare esperimenti in un universo ipotetico. [...] L'introduzione del calcolatore nella scienza è piuttosto recente, eppure la computazione costituisce già un'impostazione nuova per molti problemi. [...] La cosa forse più importante è che essa introduce nella scienza un nuovo modo di pensare; le leggi scientifiche cominciano ad essere considerate come algoritmi e molte di esse vengono studiate tramite esperimenti al calcolatore.*

[Stephen Wolfram]

La costante  $k$  del modello matematico del decadimento atomico di una massa di materiale radioattivo

$$m_{i+1} = m_i(1 - k)$$

può essere interpretata come probabilità che il singolo atomo decada nell'unità di tempo considerata (il periodo che nel modello discreto intercorre tra i due istanti successivi).

Il fenomeno della disintegrazione radioattiva può essere simulato direttamente a livello atomico: i singoli atomi possono essere infatti rappresentati mediante un vettore in cui ciascun elemento assume il valore 1 ("vivo" = non ancora disintegrato) o 0 ("morto" = già disintegrato). L'esperimento consiste nel simulare il trascorrere di un certo intervallo di tempo impiegando, per ciascuna unità di tempo della successione generata, il metodo di Monte Carlo per decidere l'eventuale decadimento di ogni singolo atomo non ancora disintegrato. Scopo dell'esperimento è la registrazione di dati statistici relativi al numero di disintegrazioni che avvengono nella singola unità di tempo; data la rarità dell'evento l'esperimento stesso dovrà essere ripetuto per un certo numero di volte al fine di rendere statisticamente rilevanti i dati registrati, di conseguenza il programma – una procedura VBA associata al pulsante di comando "Atomi" – è strutturato in tre cicli annidati:



```
Const N As Integer = 10000 ' numero iniziale di atomi
Const k As Single = 0.0005 ' costante di radioattività
Const TempEsper As Integer = 100 ' durata del singolo esperimento
Const NumEsper As Integer = 10 ' numero di esperimenti
```

```
Private Sub Atomi_Click()
Dim ATOMI(1 To 10000) As Integer
Dim tempo As Integer
Dim NumAtomi As Integer
```

```
Dim NumDis As Integer
Dim atomo As Integer
Dim esper As Integer
```

```
' ciclo di ripetizione degli esperimenti
For esper = 1 To NumEsper
  Cells(1, 2) = esper
  Randomize
  'ciclo di inizializzazione degli atomi
  For atomo = 1 To N
    ATOMI(atomo) = 1 ' "vivo"
  Next atomo
  NumAtomi = N
  Cells(2, 2) = NumAtomi

  ' ciclo di scorrimento del tempo di un esperimento
  For tempo = 0 To TempEsper
    Cells(3, 2) = tempo
    NumDis = 0

    ' ciclo di scorrimento dello spazio degli atomi
    For atomo = 1 To N
      If ATOMI(atomo) = 1 Then
        ' metodo di Monte Carlo (disintegrazione atomica)
        If Rnd() < k Then
          ATOMI(atomo) = 0 ' "morto"
          NumDis = NumDis + 1
        End If
      End If
    Next atomo

    Cells(4, 2) = NumDis
    NumAtomi = NumAtomi - NumDis
    Cells(2, 2) = NumAtomi
  Next tempo
Next Esper
End Sub
```

In esecuzione la procedura si comporta come un contatore *Geiger* virtuale visualizzando il numero di disintegrazioni registrate per ciascuna unità di tempo nella cella **B4** del foglio di lavoro di *Excel*; dovendo costruire una tabella delle frequenze della variabile aleatoria “numero di disintegrazioni nell’unità di tempo” occorre integrare il codice della procedura in modo da “contare” in un vettore (*freq*<sup>1</sup>) le frequenze prodotte dall’esperimento simulato e registrare in una variabile (*NumMaxDis*) il massimo numero di disintegrazioni osservate nell’unità di tempo:

```
Private Sub Atomi_Click()
Dim ATOMI(1 To N) As Integer
Dim FREQ(0 To N) As Integer
Dim tempo As Integer
Dim NumAtomi As Integer
Dim NumDis As Integer
```

<sup>1</sup> Il vettore *freq* è stato dimensionato al massimo numero teorico di disintegrazioni che possono avvenire nella singola unità di tempo: il numero iniziale di atomi! Per quanto teoricamente possibile, la disintegrazione di una frazione consistente di atomi in una singola unità di tempo è – data la bassa probabilità della singola disintegrazione – praticamente impossibile.

```


Dim NumMaxDis As Integer
Dim atomo As Integer
    Dim esper As Integer
    Dim dis As Integer
'ciclo di azzeramento delle frequenze
For dis = 0 To N
    FREQ(dis) = 0
Next dis
NumMaxDis = 0
' ciclo di ripetizione degli esperimenti
For esper = 1 To NumEsper
    Cells(1, 2) = esper
    Randomize
    'ciclo di inizializzazione degli atomi
    For atomo = 1 To N
        ATOMI(atomo) = 1 ' "vivo"
    Next atomo
    NumAtomi = N
    Cells(2, 2) = NumAtomi
    ' ciclo di scorrimento del tempo di un esperimento
    For tempo = 0 To TempEsper
        Cells(3, 2) = tempo
        NumDis = 0
        ' ciclo di scorrimento dello spazio degli atomi
        For atomo = 1 To N
            If ATOMI(atomo) = 1 Then
                ' metodo di Monte Carlo (disintegrazione atomica)
                If Rnd() < k Then
                    ATOMI(atomo) = 0 ' "morto"
                    NumDis = NumDis + 1
                End If
            End If
        Next atomo
        Cells(4, 2) = NumDis
        ' conteggio dell'evento (incremento della frequenza)
        FREQ(NumDis) = FREQ(NumDis) + 1
        ' eventuale aggiornamento numero massimi di disintegrazioni
        If NumDis > NumMaxDis Then NumMaxDis = NumDis
        NumAtomi = NumAtomi - NumDis
        Cells(2, 2) = NumAtomi
    Next tempo
Next esper
' ciclo di visualizzazione delle frequenze assolute e relative
For dis = 0 To NumMaxDis
    Cells(10 + dis, 1) = dis
    Cells(10 + dis, 2) = FREQ(dis)
    Cells(10 + dis, 3) = FREQ(dis) / (NumEsper * TempEsper)
Next dis
End Sub

```

Il ciclo `For ... Next` finale visualizza nel foglio di lavoro di *Excel* la tabella delle frequenze assolute e relative dei valori assunti dalla variabile aleatoria “numero di disintegrazioni nell’unità di tempo”: dato che le frequenze assolute sono calcolate sul numero complessivo di esperimenti, nel calcolo delle frequenze relative esse sono divise per il prodotto del numero degli esperimenti per il numero di unità di tempo di ciascun esperimento.

Esempio

	A	B	C
1	<b>esperimento</b>	10	
2	<b>atomi</b>	9517	
3	<b>tempo</b>	100	
4	<b>disintegrazioni</b>	5	
5			
6			
7			
8			
9	<b>disintegrazioni</b>	<b>freq. assoluta</b>	<b>freq. rel.</b>
10	0	8	0,008
11	1	39	0,039
12	2	103	0,103
13	3	141	0,141
14	4	185	0,185
15	5	159	0,159
16	6	162	0,162
17	7	96	0,096
18	8	60	0,06
19	9	34	0,034
20	10	17	0,017
21	11	4	0,004
22	12	1	0,001
23	13	1	0,001
24			
25			



**Osservazione**

Teoricamente le frequenze relative della variabile casuale “numero di disintegrazioni nell’unità di tempo” dovrebbero distribuirsi secondo la seguente legge di Poisson:

$$p(n) = \frac{\mu^n}{n!} e^{-\mu}$$

dove  $n$  è il numero di disintegrazioni,  $\mu$  è il valore medio – o “atteso” – della variabile casuale,  $e$  è una costante il cui valore è circa 2,718 e  $p(n)$  rappresenta la probabilità che si verifichino esattamente  $n$  disintegrazioni nella singola unità di tempo.


*Excel* dispone della funzione predefinita *POISSON* (categoria “Statistiche”) che calcola la probabilità che la legge di Poisson associa al primo argomento (il numero  $n$ ) fornendo come secondo argomento il valore medio della variabile casuale (la costante  $\mu$ )<sup>2</sup>. Volendo confrontare con il modello teorico di Poisson le frequenze rilevate dal programma di simulazione *Atomi* è necessario calcolare la media del “numero di disintegrazioni nell’unità di tempo” ed utilizzarne il valore come parametro  $\mu$  della legge di Poisson:

	A	B	C	D	E
1	<b>esperimento</b>	10			
2	<b>atomi</b>	9605			
3	<b>tempo</b>	100			
4	<b>disintegrazioni</b>	4			
5					
6					
7					
8					
9	<b>disintegrazioni</b>	<b>freq. assoluta</b>	<b>freq. rel.</b>		<b>prob. Teorica</b>
10	0	11	0,011	=A10*B10	=POISSON(A10; \$A\$25; FALSO)
11	1	39	0,039	=A11*B11	=POISSON(A11; \$A\$25; FALSO)
12	2	82	0,082	=A12*B12	=POISSON(A12; \$A\$25; FALSO)
13	3	155	0,155	=A13*B13	=POISSON(A13; \$A\$25; FALSO)
14	4	181	0,181	=A14*B14	=POISSON(A14; \$A\$25; FALSO)
15	5	192	0,192	=A15*B15	=POISSON(A15; \$A\$25; FALSO)
16	6	134	0,134	=A16*B16	=POISSON(A16; \$A\$25; FALSO)
17	7	99	0,099	=A17*B17	=POISSON(A17; \$A\$25; FALSO)
18	8	63	0,063	=A18*B18	=POISSON(A18; \$A\$25; FALSO)
19	9	33	0,033	=A19*B19	=POISSON(A19; \$A\$25; FALSO)
20	10	13	0,013	=A20*B20	=POISSON(A20; \$A\$25; FALSO)
21	11	5	0,005	=A21*B21	=POISSON(A21; \$A\$25; FALSO)
22	12	1	0,001	=A22*B22	=POISSON(A22; \$A\$25; FALSO)
23	13	1	0,001	=A23*B23	=POISSON(A23; \$A\$25; FALSO)
24	14	1	0,001	=A24*B24	=POISSON(A24; \$A\$25; FALSO)
25	=D25/B25	=SOMMA(B10:B24)		=SOMMA(D10:D24)	

<sup>2</sup> In realtà la funzione predefinita *POISSON* dell’ambiente *Excel* prevede tre argomenti, l’ultimo dei quali deve in questo caso essere impostato al valore “FALSO” (altrimenti, anziché la probabilità, *POISSON* calcola la probabilità cumulata).

	A	B	C	D	E
1	<b>esperimento</b>	10			
2	<b>atomi</b>	9605			
3	<b>tempo</b>	100			
4	<b>disintegrazioni</b>	4			
5					
6					
7	ATOMI				
8					
9	<b>disintegrazioni</b>	<b>freq. assoluta</b>	<b>freq. rel.</b>		<b>prob. Teorica</b>
10	0	11	0,011	0	0,008020596
11	1	39	0,039	39	0,03870533
12	2	82	0,082	164	0,093390981
13	3	155	0,155	465	0,150226944
14	4	181	0,181	724	0,181239139
15	5	192	0,192	960	0,174922686
16	6	134	0,134	804	0,140688642
17	7	99	0,099	693	0,096989596
18	8	63	0,063	504	0,058505853
19	9	33	0,033	297	0,031370465
20	10	13	0,013	130	0,015138579
21	11	5	0,005	55	0,006641353
22	12	1	0,001	12	0,002670788
23	13	1	0,001	13	0,000991426
24	14	1	0,001	14	0,00034174
25	<b>4,825742574</b>	<b>1010</b>		<b>4874</b>	

L'accordo tra probabilità teoriche e frequenze relative è confermato dal seguente grafico combinato realizzato in *Excel* selezionando al primo passaggio della *Creazione guidata Grafico* il tipo "Linee - Colonne" tra i "Tipi personalizzati":



### Esperimento

Un qualsiasi manuale di fisica riporta la seguente formula per il numero  $N$  di atomi di una massa di materiale radioattivo non ancora disintegrati all'istante  $t$  ( $N_0$  è il numero di atomi iniziale, e rappresenta il valore costante 2,71828...):

$$N(t) = N_0 e^{-kt}$$

Il numero di atomi che si disintegrano nell'intervallo di tempo  $(t_0, t_1)$  è quindi dato da

$$N(t_0) - N(t_1) = N_0 (e^{-kt_0} - e^{-kt_1})$$

e, calcolando la probabilità di disintegrazione  $p$  come rapporto tra numero di atomi disintegrati e numero totale di atomi, si ha

$$p = \frac{N(t_0) - N(t_1)}{N(t_0)} = \frac{N_0}{N(t_0)} (e^{-kt_0} - e^{-kt_1}).$$

Realizzare una funzione VBA privata Prob che restituisca il valore di  $p$  calcolato dalla formula precedente (considerare  $t_1 = t_0 + 1$ ) avendo come argomenti il numero iniziale di atomi  $N_0$ , il numero attuale di atomi  $N(t_0)$ , il tempo corrente di simulazione  $t_0$  e la costante di radioattività  $k$ . Modificare la procedura VBA Atomi\_Click presentata nel testo in modo che applichi il metodo di Monte Carlo utilizzando il valore variabile della probabilità  $p$  restituito dalla funzione Prob anziché il valore della costante di radioattività  $k$ . Ripetere in ambiente Excel gli esperimenti condotti nel testo: al livello statistico macroscopico il modello è influenzato dal cambiamento introdotto a livello microscopico?